
In SR rendezvous is accomplished by means of the
in statement. The general form of the in statement is:

in operation (formals)
     st sync_expr by sched_expr -> block
[] operation (formals)
     st sync_expr by sched_expr -> block
...
[] else -> block
ni

We have seen the synchronisation expression
already with guarded synchronised message passing.
Since the scope of the formals of the operation is the
entire guarded operation, the synchronisation and
scheduling expression can depend on the value of the
formals, and hence on the values of the arguments of the
call.

If there is no scheduling expression and several
guards are satisfied (the synchronisation expression is
true and there is a call to the operation) one of them is
chosen nondeterministically. Of the nondeterministically
chosen guard, if there are several invocations, and no
scheduling expression, the in statement services the
oldest invocation that makes the guard succeed. If there
is a scheduling expression, then the in statement services
the invocation of the guard which minimises the
scheduling expression.

Aside: SR in statement implements a more powerful
version of rendezvous than the Ada language
that only implements concurrency by
rendezvous. The reason for this is to reduce the
implementation overhead.

Example: Time Server using rendezvous

resource Time_Server
op get_time ( ) returns time:int
op delay (wake_time:int)
op tick ( ) # called by clock interrupt handler

body Time_Server
process time

var time_of_day:int := 0

do true ->
in get_time ( ) returns time ->

time := time_of_day
[] delay (wake_time)

st wake_time <= time_of_day -> skip
[] tick ( ) -> time_of_day++
ni

od
end

end

Example: Shortest Job Next Allocator using
rendezvous.

resource SJN_Allocator
op request (time:int), release ( )

body SJN_Allocator
process sjn

var free:bool := true

do true ->
in request (time) st free by time ->

free := false
[] release ( ) ->

free := true
ni

od
end

end

Example: Bounded Buffer using rendezvous

In this example the synchronisation expressions are
used to prevent any overflow or underflow occurring in
the buffer.

resource Bounded_Buffer
op deposit (item:int)
op fetch ( ) returns item:int

body Bounded_Buffer (size:int)

var buffer [1:size]:int
var count:int := 0, front:int := 0, rear:int := 0

process worker
do true ->

in deposit (item) st count < size ->
buffer [rear] := item
rear := (rear +1) mod size
count++

[] fetch ( ) returns item st count > 0 ->
item := buffer [front]
front := (front + 1) mod size
count--

ni
od

end
end

Exercises: Try to code the following in SR using
rendezvous.

1) The Readers/Writers problem. (Hint: Have the
reader and writer processes rendezvous with an
allocator process which manages access to the
database.)

2) A Centralise Dining Philosophers problem with one
waiter who controls the all the forks.


